

# CMAKE

## Introduction and best practices


---

Daniel Pfeifer

May 21, 2015

[daniel@pfeifer-mail.de](mailto:daniel@pfeifer-mail.de)

# INTEREST OVER TIME



<sup>0</sup><https://www.google.com/trends/explore#q=autoconf,automake,bjam,cmake,scons>

WHAT IS CMAKE?

---

Welcome to CMake, the **cross-platform, open-source build system**.

CMake is a **family of tools** designed to build, test and package software. CMake is used to control the software compilation process using simple **platform and compiler independent** configuration files.

CMake **generates** native makefiles and workspaces that can be used in the compiler environment of your choice.<sup>1</sup>

---

<sup>1</sup><http://www.cmake.org/>

**standalone** Make, NMake, SCons, Jom, BJam, Ninja

**integrated** Visual Studio, Xcode, Eclipse

**generators** Autotools, **CMake**, GYP

> ls zlib-1.2.8

amiga/	as400/	contrib/	doc/
examples/	msdos/	nintendods/	old/
qnx/	test/	watcom/	win32/
adler32.c	ChangeLog	<b>CMakeLists.txt</b>	compress.c
configure	crc32.c	crc32.h	deflate.c
deflate.h	FAQ	gzclose.c	gzguts.h
gzlib.c	gzread.c	gzwrite.c	INDEX
inffback.c	inffast.c	inffast.h	inffixed.h
inflate.c	inflate.h	inftrees.c	inftrees.h
make_vms.com	Makefile	Makefile.in	README
treebuild.xml	trees.c	trees.h	uncompr.c
zconf.h	zconf.h.in	zlib.3	zlib.3.pdf
zlib.h	zlib.map	zlib.pc.in	zlib2ansi
zutil.c	zutil.h		

**You** write a single configuration that CMake understands.  
**CMake** takes care that it works on all compilers and platforms.

Don't make *any* assumption about the platform or compiler!



*I was quite surprised with the speed of building Quantum GIS codebase in comparison to Autotools.<sup>2</sup>*


Task	CMake	Autotools
Configure	0:08	0:61
Make	12:15	21:16
Install	0:20	0:36
Total	12:43	22:43

---

<sup>2</sup><http://blog.qgis.org/node/16.html>

- Compiler and platform test results are cached in `CMakeCache.txt`.
- You can modify this file and re-run CMake.
- CMake also provides graphical cache editors.

# CMAKE'S GRAPHICAL CACHE EDITOR



**past** CMake generates build systems.

**today** CMake is integrated into IDEs.

**future** IDEs run CMake as a background process.<sup>3</sup>

---

<sup>3</sup><http://public.kitware.com/pipermail/cmake-developers/2015-April/025012.html>

## MODES OF OPERATION

---

---

```
1 > cmake [<options>] <path>
```

---

- If the specified path contains a CMakeCache.txt, it is treated as a build directory where the build system is reconfigured and regenerated.
- Otherwise the specified path is treated as the source directory and the current working directory is treated as build directory.

---

```
1 > mkdir build
2 > cd build
3 > time cmake ..
4 > time cmake .
5 > time make
```

---

## Issues:

- The commands `mkdir` and `time` are not portable.
- With `cmake .` you may accidentally perform an in-source configure.
- Who said that `make` is the right build command?

---

```
1 > mkdir build
2 > cd build
3 > time cmake ..
4 > time cmake .
5 > time make mylibrary
```

---

---

```
1 > cmake -E <command> [<options>...]
```

---

Run `cmake -E help` for a summary of commands.

---

```
1 > cmake -E make_directory build
2 > cmake -E chdir cmake -E time cmake ..
3 > cmake -E time cmake build
```

---



## USE CMAKE TO RUN THE BUILD SYSTEM

---

```
1 > cmake --build build \  
2         --target mylibrary \  
3         --config Release \  
4         --clean-first
```

---

Just don't:

---

```
1 > 'grep 'CMAKE_MAKE_PROGRAM:' CMakeCache.txt \  
2   | awk -F= '{print $2}'' mylibrary
```

---

---

```
1 > cat hello_world.cmake
2 foreach(greet Hello Goodbye)
3   message("${greet}, World!")
4 endforeach()
5
6 > cmake -P hello\_world.cmake
7 Hello, World!
8 Goodbye, World!
```

---

No configure or generate step is performed and the cache is not modified.

RESIST THE BASH/PERL/PYTHON TEMPTATION!

## COMMANDS FOR PROJECTS

---

Sets the minimum required version of CMake.

---

```
1 cmake_minimum_required(VERSION 3.2 FATAL_ERROR)
```

---

- Prefer the latest version of CMake.
- Please don't set 2.8 as the minimum.
- If you use 2.6 or 2.4, God kills a kitten.

Set the name and version; enable languages.

---

```
1 project(<name> VERSION <version> LANGUAGES CXX)
```

---

- CMake sets several variables based on project().
- Call to project() must be direct, not through a function/macro/include.
- CMake will add a call to project() if not found on the top level.

Embed projects as sub-projects.

---

```
1 add_subdirectory(<sourcedir> [<binarydir>])
```

---

- CMake creates a Makefile/Solution for each subproject.
- The subproject does not have to reside in a subfolder.

Make sure that all your projects can be built both **standalone** and as a **subproject** of another project:

- Don't assume that your project root is the build root.
- Don't modify global compile/link flags.
- Don't make any global changes!



Finds preinstalled dependencies

---

```
1 find_package(Qt5 REQUIRED COMPONENTS Widgets)
```

---

- Can set some variables and define imported targets.
- Supports components.

Add an executable target.

---

```
1  add_executable(tool
2      main.cpp
3      another_file.cpp
4  )
5  add_executable(my::tool ALIAS tool)
```

---

Add a library target.

---

```
1  add_library(foo STATIC
2      foo1.cpp
3      foo2.cpp
4  )
5  add_library(my::foo ALIAS foo)
```

---

- Libraries can be **STATIC**, **SHARED**, **MODULE**, or **INTERFACE**.
- Default can be controlled with **BUILD\_SHARED\_LIBS**.

- Always add namespaced aliases for libraries.
- Don't make libraries **STATIC/SHARED** unless they cannot be built otherwise.
- Leave the control of **BUILD\_SHARED\_LIBS** to your clients!

## USAGE REQUIREMENTS

---

---

```
1 target_<usage requirement>(<target>  
2   <PRIVATE|PUBLIC|INTERFACE> <lib> ...  
3   [<PRIVATE|PUBLIC|INTERFACE> <lib> ... ] ...]  
4   )
```

---

**PRIVATE** Only used for this target.

**PUBLIC** Used for this target and all targets that link against it.

**INTERFACE** Only used for targets that link against this library.

Set libraries as dependency.

---

```
1  target_link_libraries(foobar
2      PUBLIC  my::foo
3      PRIVATE my::bar
4      )
```

---

- Prefer to link against namespaced targets.
- Specify the dependencies are private or public.
- Avoid the `link_libraries()` command.
- Avoid the `link_directories()` command.
- No need to call `add_dependencies()`.



Set include directories.

---

```
1 target_include_directories(foo
2     PUBLIC include
3     PRIVATE src
4 )
```

---

- Avoid the `include_directories()` command.

Set compile definitions (preprocessor constants).

---

```
1  target_compile_definitions(foo
2      PRIVATE SRC_DIR=${Foo_SOURCE_DIR}
3      )
```

---

- Avoid the `add_definitions()` command.
- Avoid adding definitions to `CMAKE_<LANG>_FLAGS`.

Set compile options/flags.

---

```
1  if(CMAKE_COMPILER_IS_GNUCXX)
2      target_compile_options(foo
3          PUBLIC -fno-elide-constructors
4      )
5  endif()
```

---

- Wrap compiler specific options in an appropriate condition.
- Avoid the `add_compile_options()` command.
- Avoid adding options to `CMAKE_<LANG>_FLAGS`.

Set required compiler features.

---

```
1  target_compile_features(foo
2      PUBLIC
3      cxx_auto_type
4      cxx_range_for
5      PRIVATE
6      cxx_variadic_templates
7  )
```

---

- CMake will add required compile flags.
- Errors if the compiler is not supported.

- Don't add `-std=c++11` to `CMAKE_<LANG>_FLAGS`.
- Don't pass `-std=c++11` to `target_compile_options()`.



## BEST PRACTICES

---

**GOAL: NO CUSTOM VARIABLES**

---

---

```
1 set(PROJECT foobar)
2 set(LIBRARIES foo)
3
4 if(WIN32)
5     list(APPEND LIBRARIES bar)
6 endif()
7
8 target_link_libraries(${Project}
9     ${LIBRARIES}
10 )
```

---

## Hard to diagnose:

*bar* links against *foo*! But only on Windows!

---

```
1 set(PROJECT foobar)
2 set(LIBRARIES foo)
3
4 if(WIN32)
5     list(APPEND LIBRARIES bar)
6 endif()
7
8 target_link_libraries(${Project}
9     ${LIBRARIES}
10 )
```

---

## This is better:

Without variables the code is more robust and easier to read.

---

```
1 target_link_libraries(foobar PRIVATE my::foo)
2
3 if(WIN32)
4     target_link_libraries(foobar PRIVATE my::bar)
5 endif()
```

---

# DON'T USE FILE(GLOB)!

Not recommended:

---

```
1 file(GLOB sources "*.cpp")  
2 add_library(mylib ${sources})
```

---

- `file(GLOB)` is useful in scripting mode.
- Don't use it in configure mode!

## LIST ALL SOURCES EXPLICITLY

---

```
1 add_library(mylib
2   main.cpp
3   file1.cpp
4   file2.cpp
5   )
```

---

EXPLICIT IS BETTER THAN IMPLICIT



**GOAL: NO CUSTOM FUNCTIONS**

---

- I wrote many custom functions.
- Experience taught me: Most of the time it is a bad idea.

---

```
1 magic_project(<name> ...)
```

---

## WHY NO CUSTOM FUNCTIONS?

- Remember what I said about the `project()` command?
- The signature of `project()` was extended recently.

---

```
1 magic_add_library(<name>
2   SOURCES
3     <list of sources>
4   LIBRARIES
5     <list of libraries>
6   )
```

---

## WHY NO CUSTOM FUNCTIONS?

- How to add platform specific libraries?
- Harder to get started for contributors.
- Maintenance!

- Contribute it to CMake!
- If it is accepted, it is no longer a *custom function*.
- Otherwise, the reason for rejection should give you a hint.
- The maintainers are very helpful and experienced.

## SUMMARY

---



- CMake is widely used, fast, cross-platform.
- Build system generator plus platform abstraction.
- Manage build targets with usage requirements.
- Goal: No custom variables.
- Goal: No custom functions.

THANK YOU!